
Domaine Fonction Format / Explication Exemple 

Statistiques SOMME SOMME(nombre1;nombre2;...) 

Fait la somme d’un ensemble de valeurs (A1+A2+…+D2+D3) 
=SOMME(A1:D3) 

MOYENNE MOYENNE(nombre1;nombre2;...) 

Fait la moyenne d’un ensemble de valeurs  

=MOYENNE( B3:B18 )  

          Cellules B3 jusqu’à B18 

MEDIANE MEDIANE(nombre1;nombre2;...) 

Calcule la médiane d’un ensemble de valeurs  
=MEDIANE(B3:B18) 

ECARTYPEP ECARTYPEP(nombre1;nombre2;...) 

Calcule l’écart-type de l’ensemble de valeurs (population) 
=ECARTYPEP(B3:B18) 

ECARTYPE ECARTYPE(nombre1;nombre2;...) 

Calcule l’écart-type d’un ensemble de valeurs, en tenant compte d’un 

échantillon de cet ensemble.  

=ECARTYPE(B3:B18) 

MAX MAX(nombre1;nombre2;...) 

Retrouve la valeur maximale parmi un ensemble de valeurs 

 
=MAX( ListeStages!B:B ) 

colonne B de la feuille ListeStages 

MIN MIN(nombre1;nombre2;...) 

Retrouve la valeur minimale parmi un ensemble de valeurs 
=MIN(ListeStages!B:B) 

ABS ABS(nombre) 

Calcule la valeur absolue (positive) d’un nombre 

 

=ABS(A1) 

=ABS( SOMME( B3 ; B4 ; B2 )  

Valeur absolue du résultat de la somme B3+B4+B2 

Domaine Fonction Format / Explication Exemple 

ARRONDI.INF 

ARRONDI.SUP 

ARRONDI.INF ( nombre ; nb de cases ) 
ARRONDI.SUP ( nombre ; nb de cases) 

Arrondi au valeur inférieur / supérieur, en gardant le nombre de cases 

décimales indiqués   

=ARRONDI.INF(E6;0) 

=ARRONDI.SUP(E6;0)  

Arrondi la valeur de la cellule E6 vers le bas (si 16,5, on 

obtient 16) ou vers le haut (si 16,5, on obtient 17)  

ARRONDI.AU.MULTIPLE ARRONDI.AU.MULTIPLE (nombre ; multiple) 

Arrondi une valeur à la case près, selon le multiple indiqué 

=ARRONDI.AU.MULTIPLE(G2 ; 0,5) 

Arrondi au demi-point près la valeur de G2 

Tests 

 

NBVAL NBVAL(plage1 ; plage2 ...) 

Compte le nombre de cellules non vides dans une plage de cellules 
=NBVAL(A2:A7) 

NB.SI NB.SI (où voulez-vous rechercher ? ; que voulez-vous rechercher ?) 

Compte le nombre de cellules correspondant à un critère de recherche 
dans une plage de données précise 

=NB.SI(ListeStages!D:D;"Nice") 

=NB.SI( ListeStages!D:D ; B7 ) 

Cherche dans la colonne D de la feuille ListeStages la valeur 
qui est dans la cellule B7 

SI SI(condition; valeur_si_vrai; valeur_si_faux) 

Attribue à une cellule une valeur selon une condition donnée.  

=SI(C2="Oui";1;2) 

=SI( JOURS(D4;C4)<2 ; PrixKM ;  
                                        (F4-E4)*PrixKM ) 

SI.CONDITIONS 

(Office 365 / Office 2019) 

SI.CONDITIONS ( teste t1 ; valeur si t1 vrai ;  teste t2 ; valeur si t2 vrai ; … 
   VRAI ; valeur si tous sont faux ) 

Attribue à une cellule une valeur selon une condition donnée. Plusieurs 
valeurs, accompagnées de leurs conditions peuvent être indiquées.  

Les conditions sont évaluées dans l’ordre d’apparition.  

=SI.CONDITIONS(B2<9;"Non acquis"; 
                               B2<12;"En cours d'acquisition"; 

                               B2<16;"À consolider"; 
                               B2>=16;"Acquis")  

 

SIERREUR =SIERREUR (   valeur ;  valeur si erreur ) 

Attribue à la cellule soit la valeur indiquée sur le premier paramètre 

(normalement une autre fonction), soit, si celle-ci produit une erreur, la 
cellule assumera la valeur indiquée sur le deuxième paramètre.  

=SIERREUR( SOMME.SI(H3:H5;">4")/NB.SI(H3:H5;">4");  

         0) 

Si la fonction SOMME.SI(...)/NB.SI(…) fonctionne, on aura le 
résultat de celle-ci, sinon on aura 0 comme valeur.   



Domaine Fonction Format / Explication Exemple 

SOMME.SI 

(aussi disponible 
MOYENNE.SI) 

SOMME.SI (plage_à_tester ; critère ; plage_à_additionner) 

Fait la somme des valeurs contenues dans la colonne « plage à 
additionner » uniquement pour les lignes correspondant au critère  

=SOMME.SI(I3:I6;">=2";H3:H6) 

SOMME.SI.ENS  

(aussi disponible 

MOYENNE.SI.ENS, 
MAX.SI.ENS) 

SOMME.SI.ENS(plage_à_additionner ; plage_à_tester ; critère ; 
                                  plage_à_tester ; critère ; …. ) 

Fait la somme des valeurs contenues dans la colonne « plage à 
additionner » selon plusieurs critères, indiqués dans les multiples 

« plage_à_tester » avec leurs critères correspondants. 

=SOMME.SI.ENS(B2:B6; 
                              B2:B6;">=10"; B2:B6;"<15") 

(Note : on utilise dans cet exemple la même colonne pour 

faire la somme et pour les critères) 

ESTVIDE ESTVIDE (cellule) 

Vérifie si une cellule est vide ou non.  
=ESTVIDE(A2) 

ESTNUM ESTNUM (cellule) 

Vérifie si une cellule contient un numéro ou pas.  
=ESTNUM(B3) 

ESTNA ESTNA(cellule) 

Vérifie si le contenu de la cellule contient une erreur #NA. 

Retourne VRAI si la cellule contient cette erreur (ex. une fonction 
RECHERCHEV qui n’aurait pas trouvé la valeur), FAUX sinon.    

=ESTNA(C2) 

 

 

ESTERREUR ESTERREUR ( cellule ) 

Vérifie si le contenu de la cellule contient une erreur quelconque 

(#VALEUR, #NA, #DIV/0!...)  

=ESTERREUR(G7) 

Retourne VRAI si la cellule G7 contient une erreur 

NON NON (test logique) 

Inverse la valeur logique du test : si la réponse au test est VRAI, NON 
retourne FAUX ; si c’est FAUX, il retourne VRAI.  

=NON(ESTNUM(B3)) 

Retourne VRAI si la cellule B3 ne contient PAS un numéro 
(ESTNUM = VRAI) 

ET 

OU 

ET ( condition ; condition ; … )  / OU ( condition ; condition ; … ) 

Réalise un ET logique / un OU logique.  

 

Répond VRAI si toutes les conditions sont vraies (ET), ou si une seule de 
ces conditions est vraie (OU).  

Répond FAUX si une seule de ces conditions est fausse (ET), ou si toutes 
les conditions sont fausses (OU).  

=OU(B4="Ventes";H4>1000) 

=SI( OU( B4="Ventes" ; H4>1000) ; H4*1,1 ; H4 ) 

Explication : si la cellule B4 contient la valeur « ventes » ou 
si la valeur de H4 est supérieure à 1000, on calcule 10% sur 
la valeur de H4. Dans le cas contraire (ni « Ventes » sur B4, 
ni valeur supérieure à 1000), on garde la valeur de H4.   

Dates AUJOURDHUI AUJOURDHUI() 

Indique la date d’aujourd’hui. Attention à formater la cellule en format 

« date » (JJ/MM/AAAA par exemple). 

=AUJOURDHUI() 

DATE DATE ( année ; mois ; jour) 

Indique la date indiquée par les paramètres.  
=DATE(1900;1;1) 

NB.JOURS.OUVRES NB.JOURS.OUVRES ( date_début ; date_fin ; plage_dates_fériés) 

Calcule le nombre de jours ouvrés entre la date de début et la date de 

fin, en tenant compte des dates de jours fériés indiquées dans les 
cellules de la plage « dates_fériés ».  

=NB.JOURS.OUVRES( "1/5/2019"; "31/5/2019"; 
                                       E4:G4 ) 

Calcule le nombre de jours ouvrés écoulés entre 1 et 31 
mai 2019, en tenant compte des jours fériés indiqués 

dans les cellules E4 à G4.  

SERIE.JOUR.OUVRE SERIE.JOUR.OUVRE( date_début ; décalage ;  plages_dates_fériés ) 

Calcule une date x jours ouvrés (x étant la valeur décalage) après une 
date de début, en tenant compte des jours fériés indiqués dans la plage 
« dates_fériés ».  

Attention décalage doit être suppérieur à 0.  

=SERIE.JOUR.OUVRE(E15 ; 30 ; JoursFériés!A:A) 

Calcule 30 jours ouvrés après la date indiqué dans la 
cellule E15, en tenant compte des dates fériés indiquées 
dans la colonne A de la feuille « JourFériés ». 

 

MOIS.DECALER MOIS.DECALER ( date ; nb_mois) 

Décale une date de x mois.  

=MOIS.DECALER("25/12/2023";12) 

Trouve la date 12 mois après le 25/12/2023.  

FIN.MOIS FIN.MOIS( date départ ; nb de mois) 

Indique la date de la fin du nème mois (nb de mois) après la date de 
départ.  

=FIN.MOIS( AUJOURDHUI() ;0)  date fin de ce mois-ci 

=FIN.MOIS ( AUJOURDHUI() ; 1) date fin mois prochain 

=FIN.MOIS (AUJOURDHUI(); -1)   date fin mois précédent 

NO.SEMAINE NO.SEMAINE ( date ; début_semaine) 

Indique le n° de la semaine correspondant à cette date, en considérant 

que les semaines démarrent le dimanche (valeur début 1) ou le lundi 
(valeur début 2).  

=NO.SEMAINE("13/1/2019" ; 2 ) 

FRACTION.ANNEE FRACTION.ANNEE ( date_début ; date_fin) 

Calcule la proportion (nombre d’années) entre deux dates.  

=FRACTION.ANNEE("1/1/2022";"1/7/2023") 

Calcule le nb d’années écoulés entre le 1/1/22 et le 

1/7/23 ( » 1,5, c.a.d. 1 an et demi)  

JOURS JOURS ( date_fin ; date_début) 

Calcule le nombre de jours entre la date de fin et la date de début.  

Équivaut à « date_fin – date_début ».   

=JOURS("31/5/2019";"1/5/2019") 

Financières VPM VPM ( taux ; nb versements ; montant m) 

Calcule la valeur des mensualités (remboursement) d’un emprunt d’un 

montant m en n versements sur la base d’un taux d’intérêt fixe. 

=-VPM(B3/12;B4*12;B2) 

=ABS( VPM( 0,012 ; 18 ; 180000 ) ) 

VA VA ( taux ; durée ; nb versements) 

Calcule la valeur actuelle d’un emprunt (ce qu’on peut emprunter) sur 
la base d’un taux d’intérêt fixe et d’une certaine durée.  

=-VA(C3/12;C4*12;C5) 

=ABS( VA ( 0,012 ; 18 ; 900) ) 

NPM NPM ( taux ; mensualité ; montant m ) 

Calcule la durée nécessaire pour rembourser un emprunt d’un montant 
m à un taux fixe avec des versements de valeurs fixes (mensualités). 

 

=-NPM(D3;D5*12;D2) 

=ABS ( NPM ( 0,012 ; 900*12 ; 180000 ) ) 

TAUX TAUX ( durée d ; -1*mensualités ; montant) 

Calcule le taux d’intérêt pour un investissement d’un montant m, 
remboursé par des versements fixes pendant une certaine durée. 

=TAUX(E4;-1*(E5*12);E2) 

=ABS ( TAUX (18 ; -1 * (900*12) ; 180000 ) ) 

Recherche RECHERCHEV RECHERCHEV ( Critère ;  Plage données ; déplacement ;  Val proche ? ) 

Permet de rechercher une valeur (critère) dans la 1ère colonne 

(obligatoirement) d’une plage de données et de retourner une autre 
valeur située dans une colonne à droite de celle-ci (déplacement).    

=RECHERCHEV(A4;Service!B:N;B4;FAUX)  

=RECHERCHEV("0630205";Service!B:N;2;FAUX) 

On cherche la valeur « 0630205 » dans la colonne B de la 
feuille Service.  

RECHERCHEH RECHERCHEH ( Critère ;  Plage données ; déplacement ; Val proche ? ) 

Permet de parcourir la 1ère ligne d’une plage de données à la recherche 
d’une valeur (critère) et de retourner la valeur située n lignes plus bas 
(déplacement)  

=RECHERCHEH(A8;Service!B:N;B8;FAUX) 

=RECHERCHEH("Filière";Service!B:N;2;FAUX) 

On cherche la valeur « Filière » dans la première ligne 
(colonnes B à N) de la feuille Service. 

RECHERCHEX RECHERCHEX  ( Valeur recherchée ; Plage de recherche; Plage de retour ;  

                            valeur si inconnu ; type match ; type recherche )  

Permet de chercher une valeur (« valeur recherchée ») dans la « plage 

de recherche » et de retourner la valeur associée sur la « plage de 
retour ». S’il ne trouve pas la valeur, la fonction peut retourner la valeur 
#NA ou la « valeur si inconnu « (si celle-ci a été renseignée).  

 

Type match :     Type recherche : 

0  : exact     1  : vers le bas 

1  : exact ou + grand suivant   -1 : du bas vers le haut 

-1 : exact ou + petit suivant  2 : binaire vers le bas 

2 : avec * , ? , ~    -2 : binaire vers le haut 

             Type match  0 = on cherche la valeur exacte  ¯ 

=RECHERCHEX("Silver"; A1:A6 ; B1:B6 ; "inconnue"; 0 ) 

On cherche la valeur « Silver » dans les cellules A1:A6 et 
on retourne la valeur de la cellule correspondante dans 
la colonne B (entre B1 et B6). Si on ne trouve pas, la 

formule affichera la valeur « inconnue ».  

 Type match -1 = valeur exacte ou plus petite 

=RECHERCHEX( H7 ; C1:C6 ; A1:B6 ;"-" ; -1 ; 1 ) 

  Type recherche 1 : du haut vers le bas 

On cherche la valeur contenue dans la cellule H7 parmi 
les valeurs des cellules C1 à C6. On récupère les valeurs 
des colonnes A et B correspondantes (c.a.d. dans la 

même ligne de la valeur trouvée).  

EQUIV EQUIV ( valeur recherché ; plage à chercher ; mode match ) 

Permet de rechercher une valeur dans une plage et de retourner la 
position où la valeur a été trouvée (ligne si recherche dans une colonne, 

colonne si recherche dans une ligne).  

Mode match      

0 : exact  
 -1 : exact ou + petit suivant    

1 : exact ou + grand suivant 

=EQUIV("Toto"; A14:A20; 0) 

Cherche la position de la valeur « Toto » dans la plage A14 
à A20 (recherche exacte à partir de A14). 

   1 = VRAI      si la valeur de G13 se trouve dans A14:A20  

=EQUIV( 1; (A14:A20 = $G$13)*(B14:B20 = $H$13) ) 

                        et la valeur de H13 se trouve dans B14:B20 

Cherche la ligne où la colonne A contient la valeur indiquée 
sur la cellule G13 (ex. nom) et la colonne B contient la 
valeur indiquée dans H13 (ex. prénom). 

EQUIVX 

(Office 365 /  
Office 2019) 

=EQUIVX ( valeur recherché ; plage à chercher ; mode match ;  

                                                                                    mode rech ) 

Permet de rechercher une valeur dans une plage et de retourner la 

position où la valeur a été trouvée (ligne si recherche dans une colonne, 
colonne si recherche dans une ligne).  

Mode match    Mode Recherche  

0 : exact           1 : à partir de la 1ère valeur  ¯ 

-1 : exact ou + petit suivant        -1 : à partir de la dernière valeur ­ 

1 : exact ou + grand suivant 

2 : recherche avec masque ( * , ? ou ~ ) 

=EQUIVX("Ti?i"; A14:A20; 2; 1) 

Cherche la position de la valeur « Ti?i » (Titi, Tifi…) dans la 
plage A14 à A20 (recherche avec masque à partir de A14). 

 

=EQUIVX( G13 ; A14:A20 ; 0 ; -1 ) 

Cherche la valeur exacte (mode match 0) de la cellule G13 
(ex. nom) dans la colonne A mais en cherchant à partir de 
la dernière valeur, donc à partir de la cellule A20 (mode 
recherche -1)  

INDEX INDEX  ( Plage données ; ligne ; colonne ) 

Récupère le contenu de la cellule indiquée par les coordonnées ligne x 
colonne dans la plage de données indiquée. 

 

=INDEX(Service!A:N;G4;H4) 

=INDEX(Service!A:N;7;3) 



Domaine Fonction Format / Explication Exemple 

Texte GAUCHE 

DROITE 

GAUCHE ( texte ; nb de caractères à récupérer ) 
DROITE ( texte ; nb de caractères à récupérer ) 

Récupère les premiers n caractères du texte, de gauche à droite 

(GAUCHE) ou de droite à gauche (DROITE)  

=GAUCHE(C2;3) 

=DROITE(E2;3) 

STXT STXT ( texte ; position début ; nombre de caractères ) 

Récupère n caractères (« nombre de caractères ») d’un texte à partir de 
la position de début.  

=STXT(C4;11;5) 

Tata.Tartu@myco.com à @myco 
                    ^ position 11 

TROUVE TROUVE( texte recherché ; où chercher ; à partir de quelle position) 

Cherche un texte dans une cellule (p.ex., chercher un « @ » dans un 

adresse mail). Option : On peut indiquer la position à partir de laquelle 
on cherche (1 = 1ère position).   
Attention : Majuscule ≠ minuscule  

=TROUVE("@";C2;1)  

=TROUVE("@";C2) 

CHERCHE CHERCHE( texte recherché ; où chercher ; position de départ ) 

Cherche un texte dans une cellule (par exemple, chercher un « @ » 

dans un adresse mail). Option : On peut indiquer la position à partir de 
laquelle on cherche (1 = 1ère position).  

Attention : Cas insensible (majuscule = minuscule) 

=CHERCHE("@";C2;1) 

=CHERCHE("@";C2) 

SUBSTITUE SUBSTITUE (  texte ; texte recherché ; nouveau texte ; nb d’occurrences) 

Remplace le texte recherché par le nouveau texte dans la cellule 
indiquée par le paramètre texte. Répète l’opération 1 seule fois ou n 
fois si nb d’occurrences indiqué (optionnel). 

=SUBSTITUE("12,5€";"€";"") 

Remplace le « € » par rien (texte vide "") dans le texte 

« 12,5€ ».  

NBCAR NBCAR ( texte ) 

Compte le nombre de caractères contenus dans une cellule  

 

=NBCAR("C2") 

=NBCAR("texte") 

CONCAT CONTACT (texte ; texte; … ) 

Permet de concaténer (« coller ») le contenu de plusieurs 
cellules/textes/valeurs.  

=CONCAT(A2;".";B2;"@myco.com") 

JOINDRE.TEXTE JOINDRE.TEXTE( délimitateur ; ignorer si vide ; valeur ; valeur … ) 

Permet de concaténer plusieurs valeurs en les séparant par un 

délimitateur.   

=JOINDRE.TEXTE("." ; VRAI ; A2 ; B2) 

Explication : réunit les valeurs des cellules A2 et B2, en 

les séparant par un « . ». 

EPURAGE EPURAGE ( texte ) 

Élimine certains caractères spéciaux non visibles à l’œil nu.  
=EPURAGE(D13) 

SUPPRESPACE SUPPRESPACE ( texte ) 

Supprime les espaces blancs indésirables au début, à la fin et au milieu 
du texte (si plusieurs espaces blancs de suite) 

=SUPPRESPACE(D13) 

 

TEXTE TEXTE ( valeur ; format ) 

Convertit une valeur numérique en texte suivant le format indiqué. 

Exemples de format :  

"JJJJ"   à jour de la semaine (la valeur doit être une date) 
"mmm" à mois de l’année en abrégé (la valeur doit être une date) 

"0,0 %" à affiche la valeur avec un pourcentage   

 

=TEXTE("1/1/2024";"MMM")  

Affiche le nom du mois en abrégé, ici « janv »  

MAJUSCULE 

MINUSCULE 

NOMPROPRE 

MAJUSCULE ( texte ),  
MINUSCULE ( texte ) 

NOMPROPRE ( texte ) 

Convertit le texte en majuscules, minuscules ou 1ère majuscule  

 

=MAJUSCULE ( "ToTo" ) à  TOTO 

=MINUSCULE ( "ToTo" ) à toto 

=NOMPROPRE ("ToTo tutu" ) à Toto Tutu 

Conversion VALEURNOMBRE VALEURNOMBRE(  texte ; séparateur décimal ) 

Convertit le texte en valeur numérique en tenant compte du séparateur 
décimal indiqué (« . » ou « , »), #VALEUR si le texte ne correspond pas à 

une valeur numérique.  

=VALEURNOMBRE("12.25" ; ".")    à   12,25 

CNUM CNUM ( texte ) 

Convertit le texte en valeur numérique (ou date). Si le texte ne 
représente pas une valeur numérique, on aura l’erreur #VALEUR. 

=CNUM("1/1/2025")  

transforme le texte « 1/1/2025 » en date 1/1/2025 

=CNUM(A22) à transforme le 

contenu de la cellule A22 en valeur numérique 

DATEVAL DATEVAL( texte ) 

Convertit une texte contenant une date en valeur date. Si le texte ne 
contient pas une date, DATEVAL produit l’erreur #VALEUR.  

=DATEVAL("1/1/2025")  à 1/1/2025 

Divers  UNIQUE UNIQUE( plage données ) 

Renvoie une liste de valeurs uniques au sein de plage. Si une même 

valeur apparait plusieurs fois dans la plage de données, elle 
n’apparaitra qu’une seule fois sur la liste générée par UNIQUE.  

=UNIQUE(C2:C127) 

DECALER DECALER(  cellule de départ ; décalage lignes ; décalage colonnes ;  
hauteur ; largeur ) 

Définit une « vue » sur les données, c.a.d. une plage dynamique définie 

à partir de la cellule de départ avec une hauteur (i.e. nombre de lignes) 
et une largeur (i.e. nombre de colonnes) précise.  

 

  point de départ 

=DECALER( Service!$A$1 ;  
 0 ; 0 ;             on se décale 0 lignes et 0 cols du départ  

 NBVAL(Service!$A:$A) ;          nb de lignes à prendre 

 NBVAL(Service!$1:$1) )             nb de colonnes  

Recherche BD BDLIRE BDLIRE ( Plage données ;  champ ; plage critères ) 

Récupère le contenu de la colonne champ pour la cellule correspondant 
aux critères indiqués dans la plage de critères.  

Affiche l’erreur #NOMBRE! si plusieurs cellules correspondent aux 
critères.  

Attention : la plage de critères doit contenir les noms des colonnes et 
des valeurs pour certaines de ces colonnes. 

=BDLIRE(Service!A:N;"Niveau";13:14) 

=BDLIRE(Service!A:N;H13;A13:N14) 

BDNBVAL BDNBVAL ( plage données ; champ ; plage critères) 

Compte le nombre de cellules non vides contenues dans un champ 

précis (champ) de la plage de données qui respectent les critères 
indiqués dans la plage de critères.  

Attention : la plage de critères doit contenir les noms des colonnes et 

des valeurs pour certaines de ces colonnes. 

=BDNBVAL(Service!A:N;"Matière";13:14) 

=BDNBVAL(Service!A:N;C13;A13:N14) 

 

BDMOYENNE BDMOYENNE ( Plage données; champ ; plage critères) 

Calcule la moyenne des valeurs contenues dans la colonne champ pour 
les cellules qui respectent les critères indiqués dans la plage de critères. 

Attention : la plage de critères doit contenir les noms des colonnes et 
des valeurs pour certaines de ces colonnes. 

=BDMOYENNE(Service!A:N;"eqTD";13:15) 

=BDMOYENNE(Service!A:N;N13;A13:N15) 

BDSOMME BDSOMME ( Plage données ;  champ ; plage critères ) 

Réalise la somme des valeurs contenues dans la colonne champ de la 
plage de données pour les cellules qui correspondent aux critères 

indiqués dans la plage de critères. Attention : celle-ci doit contenir les 
noms des colonnes et des valeurs recherchées. 

 

=BDSOMME(Service!A:N;"eqTD";13:14) 

=BDSOMME(Service!A:N;N13;A13:N14) 

BDMAX BDMAX ( Plage données ;  champ ; plage critères) 

Retrouve la valeur maximale contenue dans la colonne champ pour les 
cellules de la plage de données respectant les critères indiqués dans la 

plage de critères. Attention : celle-ci doit contenir les noms des colonnes 
et des valeurs pour certaines de ces colonnes. 

=BDMAX(Service!A:N;"Volume";13:14) 

=BDMAX(Service!A:N;M13;A13:A14) 

BDMIN BDMIN ( Plage données ;  champ ; plage critères) 

Retrouve la valeur minimale contenue dans la colonne champ pour les 

cellules de la plage de données respectant les critères indiqués dans la 
plage de critères. Attention : celle-ci doit contenir les noms des colonnes 

et des valeurs pour certaines de ces colonnes. 

=BDMIN(Service!A:N;"Volume";13:14) 

=BDMIN(Service!A:N;M13;A13:A14) 

 


